

**FORSCHUNGSZENTRUM JÜLICH GmbH**  
**Zentralinstitut für Angewandte Mathematik**  
**D-52425 Jülich, Tel. (02461) 61-6402**

Interner Bericht

**On Solving the McConnell Equations  
in Biochemistry**

*Johannes Grotendorst, Paul Jansen,  
Siegfried M. Schoberth\**

KFA-ZAM-IB-9314

Oktober 1993  
(Stand 02.03.94)

(\* Institut für Biotechnologie 1, Forschungszentrum Jülich

This article will be published in the Maple Technical Newsletter, Issue 11.



# On Solving the McConnell Equations in Biochemistry

Johannes Grotendorst and Paul Jansen

Central Institute for Applied Mathematics  
Research Centre Juelich  
D-52425 Juelich  
Germany

Siegfried M. Schoberth

Institute for Biotechnology 1  
Research Centre Juelich  
D-52425 Juelich  
Germany

In recent years, in vivo nuclear magnetic resonance (NMR) spectroscopy has allowed to measure rate constants of transport and diffusion across living cell membranes [1-2]. One of the authors (S.M.S.) is interested in a special NMR technique -inversion transfer- to study bacterial systems [3]. The theoretical basis for the analysis of inversion transfer experiments is a system of differential equations first formulated by McConnell [4]. These equations describe the rate of change of nuclear spin magnetization of a single nuclear species which is transferred back and forth between two different magnetic environments (A, B) by kinetic processes. In this article we show, how the McConnell equations, a linear inhomogeneous system of differential equations with constant coefficients, can be solved elegantly by using symbolic matrix algebra only, in particular by using matrix exponentials. This method utilizes the special structure of the ODE system and therefore is faster and more direct than simply using `dsolve`, the solver for differential equations in Maple. It is also an independent method compared to the solution techniques found in the literature [5-7]. In addition, methods of mixed symbolic-numeric type for the determination of the formal parameters involved in the analytical solutions are described.

This application is presented as a Maple worksheet, a new feature of the graphical user interface in Maple V Release 2, for creating mathematically live documents. It combines Maple input, output, text and graphics in one easily accessible document. The goal of this worksheet is to illustrate the valuable assistance of symbolic computation in modeling and solving a special mathematical problem in Biochemistry.

The McConnell equations as formulated by Led and Gesmar [5] to study chemical exchange rates are given as follows

```
> eq1 := diff(MA(t), t) = -k[A1]*MA(t) + k[-1]*MB(t) + k[A];  
eq2 := diff(MB(t), t) = k[1]*MA(t) - k[B1]*MB(t) + k[B];
```

$$eq1 := \frac{\partial}{\partial t} MA(t) = -k_{[A1]} MA(t) + k_{[-1]} MB(t) + k_{[A]}$$

$$eq2 := \frac{\partial}{\partial t} MB(t) = k_{[1]} MA(t) - k_{[B1]} MB(t) + k_{[B]}$$

Here, MA(t) and MB(t) describe the time-dependent peak heights of the magnetic resonance signals in A and B, respectively. These signals are measured in NMR experiments. The parameters k[1] and k[-1], interpreted as indexed names in Maple, characterize the first-order rate constants for the forward and reverse reactions. The meaning of the other parameters will become clear in the sequel. From linear algebra we know that linear systems of differential equations with constant coefficients can be solved with the help of matrix exponentials [8]. To do it in Maple we first have to define the appropriate matrices.

```
> with(linalg): #load the linear algebra package
  A := matrix([[ -k[A1], k[-1]], [k[1], -k[B1]]]);
  C := vector([k[A], k[B]]);
  Y0 := vector([MA0, MB0]);
```

Warning: new definition for norm

Warning: new definition for trace

$$A := \begin{bmatrix} -k_{[A1]} & k_{[-1]} \\ k_{[1]} & -k_{[B1]} \end{bmatrix}$$

$$C := \begin{bmatrix} k_{[A]} & k_{[B]} \end{bmatrix}$$

$$Y0 := [MA0 \ MB0]$$

Applying the Maple function for the evaluation of matrix exponentials leads to the following solution of the inhomogeneous ODE system:

```
> F := multiply(exponential(A,t), Y0):
  f := s -> multiply(exponential(A, t-s), C);
  P := [Int(f(s)[1], s=0..t), Int(f(s)[2], s=0..t)];
```

$$f := s \rightarrow \text{multiply}(\text{exponential}(A, t - s), C)$$

$$P := \left[ \int_0^t -\frac{1}{2} \left( -k_{[A]} \sqrt{\%1} - k_{[A]} k_{[A1]} \%2 + k_{[A]} k_{[A1]} \%3 + k_{[A]} k_{[B1]} \%2 - k_{[A]} k_{[B1]} \%3 \right. \right.$$

$$\left. - k_{[A]} \sqrt{\%1} \%2 + 2 k_{[-1]} k_{[B]} \%2 - 2 k_{[-1]} k_{[B]} \%3 \right) / \sqrt{\%1} ds, \int_0^t \frac{1}{2} \left( -2 k_{[1]} k_{[A]} \%2 \right.$$

$$\left. + 2 k_{[1]} k_{[A]} \%3 + k_{[B]} \%3 \sqrt{\%1} + k_{[B]} k_{[B1]} \%2 - k_{[B]} k_{[B1]} \%3 - k_{[B]} k_{[A1]} \%2 \right.$$

$$\left. + k_{[B]} k_{[A1]} \%3 + k_{[B]} \sqrt{\%1} \%2 \right) / \sqrt{\%1} ds \Bigg]$$

$$\%1 := k_{[A1]}^2 - 2 k_{[A1]} k_{[B1]} + k_{[B1]}^2 + 4 k_{[-1]} k_{[1]}$$

$$\%2 := e^{-1/2(k_{1A1} + k_{1B1} + \sqrt{\%1})(t-s)}$$

$$\%3 := e^{-1/2(k_{1A1} + k_{1B1} - \sqrt{\%1})(t-s)}$$

Here, the % label signifies common subexpressions in the output, a way Maple makes output more compact. The matrix exponential of a  $d \times d$  square matrix  $A$  is defined by

$$\exp(A) = I + A + 1/2!A^2 + 1/3!A^3 + \dots + 1/n!A^n + \dots,$$

where  $I$  is the identity matrix and the matrix series on the right hand side can be considered as  $d^2$  (scalar) series, one for each of the elements of  $\exp(A)$  [8]. Now, the solution of the inhomogeneous ODE system is obtained by adding the solution  $F$  of the corresponding homogeneous ODE system and the particular solution  $P$ .

```
> S1 := add(F, map(value, P));
```

It should be noted that the solution of the equivalent one dimensional ODE is found in a formal identical manner. The correctness of the one dimensional solution is easily verified.

```
> exp(a*t)*y0 + int(exp(a*(t-s))*c, s=0..t):
  evalb(expand(diff(" , t) = a*" + c));
```

*true*

If matrix  $A$  is invertible then the solution of the inhomogeneous ODE system can be computed with matrix algebra only. We have

```
> AC := multiply(inverse(A), C):
  S2 := add(multiply(exponential(A, t), add(Y0, AC)), -AC):
```

Again, this direct solution can be motivated by the one dimensional case:

```
> dsolve({diff(y(t), t) = a*y(t) + c, y(0) = y0}, y(t));
```

$$y(t) = -\frac{c}{a} + e^{at} \left( \frac{c}{a} + y0 \right)$$

Comparing the solutions  $S1$  and  $S2$  yields

```
> map(simplify, add(S1, -S2));
```

[0 0]

Next, we introduce physical boundary conditions to obtain a special solution. Let  $MeA$  and  $MeB$  denote the unperturbed equilibrium magnetizations of  $A$  and  $B$  respectively, i.e. the limits of  $MA(t)$  and  $MB(t)$  for  $t \rightarrow +\infty$ . These limits are given by the components of the constant vector  $-AC$  in solution  $S2$ , provided that both eigenvalues of  $A$  are negative.

> MeA = -AC[1], MeB = -AC[2]; #(\*)

$$MeA = \frac{k_{[A]} k_{[BI]} + k_{[-1]} k_{[B]}}{k_{[AI]} k_{[BI]} - k_{[-1]} k_{[1]}}, MeB = \frac{k_{[A]} k_{[1]} + k_{[B]} k_{[AI]}}{k_{[AI]} k_{[BI]} - k_{[-1]} k_{[1]}}$$

Solving for k[A] and k[B] yields

> solve({}, {k[A], k[B]});  
assign("):

$$\left\{ k_{[A]} = -k_{[-1]} MeB + k_{[AI]} MeA, k_{[B]} = MeB k_{[BI]} - k_{[1]} MeA \right\}$$

Inserting these expressions into the system of ODEs and collecting coefficients results in the following system of equations:

> eq1 := collect (eq1, [k[A1], k[-1]]);  
eq2 := collect (eq2, [k[B1], k[1]]);

$$eq1 := \frac{\partial}{\partial t} MA(t) = (-MA(t) + MeA) k_{[AI]} + (MB(t) - MeB) k_{[-1]}$$

$$eq2 := \frac{\partial}{\partial t} MB(t) = (-MB(t) + MeB) k_{[BI]} + (MA(t) - MeA) k_{[1]}$$

Now, the inhomogeneous ODE system is transformed to a homogeneous system by MA(t) -> MA1(t) + MeA and MB(t) -> MB1(t) + MeB.

> neweq1 := eval(subs(MA(t)=MA1(t)+MeA,  
MB(t)=MB1(t)+MeB, eq1)); #(\*\*)  
neweq2 := eval(subs(MA(t)=MA1(t)+MeA,  
MB(t)=MB1(t)+MeB, eq2));

$$neweq1 := \frac{\partial}{\partial t} MA1(t) = -MA1(t) k_{[AI]} + MB1(t) k_{[-1]}$$

$$neweq2 := \frac{\partial}{\partial t} MB1(t) = -MB1(t) k_{[BI]} + MA1(t) k_{[1]}$$

For the solution of this system we use matrix exponentials again.

```
> time1 := time();
Y1 := vector([MA10, MB10]);
funsys := multiply(exponential(A,t), Y1);
# adding the boundary values
linsol := add(funsys, vector([MeA, MeB]));
time1 := time() - time1;
```

$$\begin{aligned}
linsol := & \left[ \frac{1}{2} \left( MA10 \%3 \sqrt{\%1} + MA10 k_{[AI]} \%2 - MA10 k_{[AI]} \%3 - MA10 k_{[BI]} \%2 \right. \right. \\
& + MA10 k_{[BI]} \%3 + MA10 \sqrt{\%1} \%2 - 2 k_{[-1]} MB10 \%2 + 2 k_{[-1]} MB10 \%3 \left. \right) / \sqrt{\%1} \\
& + MeA - \frac{1}{2} \left( 2 k_{[1]} MA10 \%2 - 2 k_{[1]} MA10 \%3 - MB10 \%3 \sqrt{\%1} - MB10 k_{[BI]} \%2 \right. \\
& + MB10 k_{[BI]} \%3 + MB10 k_{[AI]} \%2 - MB10 k_{[AI]} \%3 - MB10 \sqrt{\%1} \%2 \left. \right) / \sqrt{\%1} \\
& \left. + MeB \right]
\end{aligned}$$

$$\%1 := k_{[AI]}^2 - 2 k_{[AI]} k_{[BI]} + k_{[BI]}^2 + 4 k_{[-1]} k_{[1]}$$

$$\%2 := e^{-1/2 (k_{[AI]} + k_{[BI]} + \sqrt{\%1}) t}$$

$$\%3 := e^{-1/2 (k_{[AI]} + k_{[BI]} - \sqrt{\%1}) t}$$

$$time1 := .330$$

Thus, under condition (\*) the solution of the original inhomogeneous ODE system is simplified to a sum where one term is the solution of the homogeneous ODE system (\*\*) and the other term, usually the particular solution, is constant. Applying the collect function recursively to each component of the vector linsol provides a compact analytical solution of the following general form:

$$\begin{aligned}
MA(t) &= C1 * \exp(lam1 * t) + C2 * \exp(lam2 * t) + MeA \\
MB(t) &= C3 * \exp(lam1 * t) + C4 * \exp(lam2 * t) + MeB
\end{aligned}$$

> linsol := map(collect, linsol, [%2, %3, MA10, MB10]);

$$\begin{aligned}
linsol := & \left[ \left( \frac{1}{2} \frac{(k_{[AI]} - k_{[BI]} + \sqrt{\%1}) MA10}{\sqrt{\%1}} - \frac{k_{[-1]} MB10}{\sqrt{\%1}} \right) e^{-1/2 (k_{[AI]} + k_{[BI]} + \sqrt{\%1}) t} \right. \\
& + \left( \frac{1}{2} \frac{(\sqrt{\%1} - k_{[AI]} + k_{[BI]}) MA10}{\sqrt{\%1}} + \frac{k_{[-1]} MB10}{\sqrt{\%1}} \right) e^{-1/2 (k_{[AI]} + k_{[BI]} - \sqrt{\%1}) t} + MeA \\
& \left( -\frac{k_{[1]} MA10}{\sqrt{\%1}} - \frac{1}{2} \frac{(-k_{[BI]} + k_{[AI]} - \sqrt{\%1}) MB10}{\sqrt{\%1}} \right) e^{-1/2 (k_{[AI]} + k_{[BI]} + \sqrt{\%1}) t} \\
& \left. + \left( \frac{k_{[1]} MA10}{\sqrt{\%1}} - \frac{1}{2} \frac{(k_{[BI]} - \sqrt{\%1} - k_{[AI]}) MB10}{\sqrt{\%1}} \right) e^{-1/2 (k_{[AI]} + k_{[BI]} - \sqrt{\%1}) t} + MeB \right]
\end{aligned}$$

$$\%1 := k_{[AI]}^2 - 2 k_{[AI]} k_{[BI]} + k_{[BI]}^2 + 4 k_{[-1]} k_{[1]}$$

It is easily verified that the parameters lam1 and lam2 are the eigenvalues of the coefficient matrix of the homogeneous ODE system. We have

```
> eigenvals(A);
```

$$-\frac{1}{2}k_{[AI]} - \frac{1}{2}k_{[BI]} + \frac{1}{2}\sqrt{k_{[AI]}^2 - 2k_{[AI]}k_{[BI]} + k_{[BI]}^2 + 4k_{[-1]}k_{[1]}},$$

$$-\frac{1}{2}k_{[AI]} - \frac{1}{2}k_{[BI]} - \frac{1}{2}\sqrt{k_{[AI]}^2 - 2k_{[AI]}k_{[BI]} + k_{[BI]}^2 + 4k_{[-1]}k_{[1]}}$$

Next, we solve the homogeneous ODE system (\*\*) by applying the Maple function dsolve, the solver for differential equations in Maple.

```
> time2 := time():
start := MA1(0) = MA10, MB1(0) = MB10:
fcns := {MA1(t), MB1(t)}:
dsol := dsolve({neweq1, neweq2, start}, fcns):
assign(dsol):
#adding the boundary values
dsol := vector([MA1(t) + MeA, MB1(t) + MeB]):
time2 := time() - time2:
speedup := time2/timel:
```

*time2 := 22.760*

*speedup := 68.96969697*

Thus, dsolve needs much more time to find the solution compared to the method using the matrix exponentials. This is explained by the more general algorithms used in dsolve which work for nonlinear equations and may not be efficient for homogeneous linear systems. The time (in seconds) was measured on a RS/6000-32H workstation. Comparing the result of dsolve with that found by using matrix exponentials yields

```
> map(simplify, add(dsol, -linsol));
```

[0 0]

If we define the parameters

```
> MA10 := MA0 - MeA: MB10 := MB0 - MeB:
k[A1] := 1/T[A1] + k[1]: k[B1] := 1/T[B1] + k[-1]:
```

we get analytical solutions for MA(t) and MB(t) which depend on the eight parameters MA0, MB0, MeA, MeB, T[A1], T[B1], k[1], k[-1]. Here, 1/T[A1] and 1/T[B1] denote the relaxation rates of the spins in the two sites, whereas MA0 and MB0 denote the initial values of MA(t) and MB(t) at time 0. Usually, the parameters involved in MA(t) and MB(t) are determined by a nonlinear least-squares analysis, i.e. by fitting the model parameters to experimental data for MA(t) and MB(t) obtained at different values of time [5-7]. The symbolic evaluation of the Jacobian matrix, required for the numerical fit program, is readily done with the help of the Maple procedure jacobian. For the translation into optimized FORTRAN code and the generation of a driver program for the fitting routine we used Macrofort [9], a Maple package for FORTRAN code generation. The numerical parameter-fitting itself is accomplished by the ACM algorithm NL2SOL [10] which is based on the Levenberg-Marquardt algorithm and which needs analytic Jacobian matrices as input. The Macrofort program is capable to construct complete and ready to compile FORTRAN code for a given set of functions and parameters.

Now, for plotting the functions MA(t) and MB(t) we substitute special fitted values obtained from NMR data to investigate transport processes in a special biological system [3].

```
> k[1] := 6.6: k[-1] := 10.5: MeA := 153.5: MeB := 78.8:
  MA0 := 130.4: MB0 := 27.08: T[A1] := 1.4: T[B1] := 1.8:
```

The parameter values lead to the following negative eigenvalues of matrix A:

```
> eigenvals(A);
```


-.652671519, -17.71716975

The given boundary conditions of MA(t) and MB(t) for  $t \rightarrow +\infty$  are reproduced by

```
> map(limit, linsol, t=+infinity);
```

[153.5000000 78.8000000]

```
> with(plots):
  p1 := plot({seq(linsol[i], i=1..2)}, t=0..5):
  t1 := textplot([4, 79, 'MB(t)'], align=ABOVE):
  t2 := textplot([4, 147, 'MA(t)']):
  display({p1, t1, t2});
```



### Acknowledgement

The authors would like to thank Edgar Wermuth for his help to find the explicit solution in terms of matrix exponentials. Valuable comments of the referee for improving the readability of the article are acknowledged.

## References

- [1] K. Kirk, *NMR Biomed.* 3, 1-16, (1990).
- [2] P. W. Kuchel, *NMR Biomed.* 3, 102-119, (1990).
- [3] S. M. Schoberth, A. A. de Graaf, B. E. Chapman, P. W. Kuchel, and H. Sahn, XV International Conference on Magnetic Resonance in Biological Systems, Jerusalem, Israel, August 16-21, (1993).
- [4] H. M. McConnell, *J. Chem. Phys.* 28, 430-431, (1958).
- [5] J. J. Led and H. Gesmar, *J. Magn. Res.* 49, 444-464, (1982).
- [6] P. W. Kuchel and B. E. Chapman, *J. Theor. Biol.* 105, 569-589, (1983).
- [7] G. Robinson, B. E. Chapman, and P. W. Kuchel, *Eur. J. Biochem.* 143, 643-649, (1984).
- [8] P. Hartman, "Ordinary Differential Equations", John Wiley & Sons, Inc., (1964).
- [9] C. Gomez, "Macrofort: a FORTRAN code generator in Maple", Maple Share Library.
- [10] J. E. Dennis, D. M. Gay, and R. E. Welsch, *ACM Trans. Math. Software*, Vol. 7, No. 3, (1981).